

# INSTRUCTION MANUAL



**Model 257 and 257-L**  
**(Watermark 200)**  
**Soil Moisture Sensor**

Revision: 3/05

Copyright (c) 1993-2005  
Campbell Scientific, Inc.

# **Warranty and Assistance**

---

The **MODEL 257 AND 257-L (WATERMARK 200) SOIL MOISTURE SENSOR** is warranted by CAMPBELL SCIENTIFIC, INC. to be free from defects in materials and workmanship under normal use and service for twelve (12) months from date of shipment unless specified otherwise. Batteries have no warranty. CAMPBELL SCIENTIFIC, INC.'s obligation under this warranty is limited to repairing or replacing (at CAMPBELL SCIENTIFIC, INC.'s option) defective products. The customer shall assume all costs of removing, reinstalling, and shipping defective products to CAMPBELL SCIENTIFIC, INC. CAMPBELL SCIENTIFIC, INC. will return such products by surface carrier prepaid. This warranty shall not apply to any CAMPBELL SCIENTIFIC, INC. products which have been subjected to modification, misuse, neglect, accidents of nature, or shipping damage. This warranty is in lieu of all other warranties, expressed or implied, including warranties of merchantability or fitness for a particular purpose. CAMPBELL SCIENTIFIC, INC. is not liable for special, indirect, incidental, or consequential damages.

Products may not be returned without prior authorization. The following contact information is for US and International customers residing in countries served by Campbell Scientific, Inc. directly. Affiliate companies handle repairs for customers within their territories. Please visit [www.campbellsci.com](http://www.campbellsci.com) to determine which Campbell Scientific company serves your country. To obtain a Returned Materials Authorization (RMA), contact CAMPBELL SCIENTIFIC, INC., phone (435) 753-2342. After an applications engineer determines the nature of the problem, an RMA number will be issued. Please write this number clearly on the outside of the shipping container. CAMPBELL SCIENTIFIC's shipping address is:

**CAMPBELL SCIENTIFIC, INC.**

RMA# \_\_\_\_\_

815 West 1800 North

Logan, Utah 84321-1784

CAMPBELL SCIENTIFIC, INC. does not accept collect calls.

# 257 Table of Contents

---

*PDF viewers note: These page numbers refer to the printed version of this document. Use the Adobe Acrobat® bookmarks tab for links to specific sections.*

<b>1. General Description</b> .....	1
<b>2. Specifications</b> .....	2
<b>3. Installation and Removal</b> .....	2
<b>4. Wiring</b> .....	3
<b>5. Description of Measurement</b> .....	4
5.1 Edlog Dataloggers.....	4
5.1.1 Program Instruction 5.....	4
5.1.2 Program Instruction 59.....	5
5.2 CRBasic Dataloggers.....	5
5.2.1 BRHalf Instruction.....	5
5.3 Calculate Soil Water Potential.....	5
5.3.1 Linear Relationship.....	6
5.3.2 Non-Linear Relationship.....	6
<b>6. Example Programs</b> .....	7
6.1 Program Examples — CR10X.....	7
6.1.1 Program Example — CR10X Linear Relationship .....	7
6.1.2 Program Example — CR10X Non-Linear Relationship .....	9
6.2 Program Example — CR1000 .....	12
<b>7. Interpreting Results</b> .....	13
<b>8. Troubleshooting</b> .....	13
<b>9. Reference</b> .....	13

## **Figures**

1-1. 257 Soil Moisture Sensor .....	2
4-1. 257 Schematic .....	3

## **Tables**

4-1. Wiring for Edlog (CR10X) Programming Example .....	4
4-2. Wiring for CRBasic (CR1000) Programming Example.....	4
5-1. Excitation and Voltage Ranges for Edlog Dataloggers.....	5
5-2. Excitation and Voltage Ranges for CRBasic Dataloggers .....	5
5-3. Comparison of Estimated Soil Water Potential and $R_s$ at 21°C .....	7

# ***Model 257 and 257-L (Watermark 200) Soil Moisture Sensor***

---

## **1. General Description**

The 257 (Watermark 200) soil moisture sensor provides a convenient method of estimating water potential between 0 and 200 kPa (wetter soils) with a Campbell Scientific datalogger. Supported dataloggers include the 21X, CR1000, CR10(X), CR23X, CR510, and CR7. Models 257 and 257-L connect directly to a datalogger. For applications requiring many sensors on a multiplexer, or for soil water matric potential measurements with the CR200-series dataloggers, the 253 and 253-L soil moisture sensor must be used.

The -L option on the model 257 Soil Moisture Sensor (257-L) indicates that the cable length is user specified. Otherwise, the 257 is shipped with 25 feet of cable. This manual refers to both the 257 and 257-L as the 257.

The Watermark block estimates water potential. For applications requiring high accuracy, call a Campbell Scientific applications engineer for information on precision soil moisture measurement systems.

The Watermark consists of two concentric electrodes embedded in a reference granular matrix material. The granular matrix material is surrounded by a synthetic membrane for protection against deterioration. An internal gypsum tablet buffers against the salinity levels found in irrigated soils.

If cultivation practices allow, the sensor can be left in the soil all year, eliminating the need to remove the sensor during the winter months.

---

### **NOTE**

The black outer jacket of the cable is Santoprene<sup>®</sup> rubber. This compound was chosen for its resistance to temperature extremes, moisture, and UV degradation. However, this jacket will support combustion in air. It is rated as slow burning when tested according to U.L. 94 H.B. and will pass FMVSS302. Local fire codes may preclude its use inside buildings.

---




FIGURE 1-1. 257 Soil Moisture Sensor

## 2. Specifications

Range:	0 to 200 kPa
Dimensions:	8.26 cm (3.25") long with a 1.91 cm (0.75") diameter
Weight:	363 g (0.8 lbs)

## 3. Installation and Removal

Placement of the 257 is important. To acquire representative measurements, avoid high spots, slope changes, or depressions where water puddles. Typically, the sensor must be located in the root system of the crop.

1. Soak the sensors overnight in irrigation water. Always install a wet sensor. If time permits, allow the sensor to dry for 1 to 2 days after soaking, and repeat the soak/dry cycle twice to improve sensor response
2. Make a sensor access hole to the depth required with a 22 mm (7/8") diameter rod. Fill the hole with water and push the sensor to the bottom of the hole. Very coarse or gravelly soils may require an oversized hole (25 to 32 mm) to prevent abrasion damage to the sensor membrane. In this case, you will need to "grout in" the sensor with a slurry made from the sample soil to get a snug fit in the soil.

Snug fit in the soil is most important. Lack of a snug fit is the premier problem in sensor effectiveness. In gravelly soils, and with deeper

sensors, sometimes it is hard to get the sensor in without damaging the membrane. The ideal method of making the access hole is to have a “stepped” tool that makes an oversized hole for the upper portion and an exact size hole for the lower portion. In either case, the hole needs to be carefully backfilled and tamped down to prevent air pockets within could allow water to channel down to the sensor.

A length of ½” class 315 PVC pipe fits snugly over the sensor collar and can be used to push in the sensor.

You can leave the PVC in place with the wires threaded through the pipe and the open end taped shut (duct tape is adequate). This practice also makes it easy to remove sensors used in annual crops. When doing this, solvent weld the PVC pipe to the sensor collar. Use PVC/ABS cement on the stainless steel sensors with the green top. Use clear PVC cement only on the PVC sensors with the gray top.

3. When removing sensors prior to harvest in annual crops, do so just after the last irrigation when the soil is moist. Do not pull the sensor out by the wires. Careful removal prevents sensor and membrane damage.
4. When sensors are removed for winter storage, clean, dry, and place them in a plastic bag.

## 4. Wiring

The 257 wiring diagram is illustrated in Figure 4-1. The red lead (Positive Signal) is inserted into any single-ended analog channel, the black lead into any excitation channel, and the white lead (Negative Signal) to any Analog Ground (CR10(X), CR510) or Ground (21X, CR23X, CR1000, CR7).

Installed in the cable is a capacitor circuit which blocks galvanic action due to the differences in potential between the datalogger earth ground and the electrodes in the block. Such current flow would cause rapid block deterioration.




FIGURE 4-1. 257 Schematic

Wiring for the following program examples is shown in Table 4-1 and Table 4-2.

<b>TABLE 4-1. Wiring for Edlog (CR10X) Programming Example</b>			
<b>Sensor</b>	<b>Wire</b>	<b>Function</b>	<b>Channel</b>
<b>107</b>	Black	Excitation	E1
	Red	Positive Signal	SE1 (1H)
	Purple	Negative Signal	AG
	Clear	Shield	G
<b>257</b>	Black	Excitation	E2
	Red	Positive Signal	SE2 (1L)
	White	Negative Signal	AG
	Clear	Shield	G

<b>TABLE 4-2. Wiring for CRBasic (CR1000) Programming Example</b>			
<b>Sensor</b>	<b>Wire</b>	<b>Function</b>	<b>Channel</b>
<b>107</b>	Black	Excitation	EX1
	Red	Positive Signal	SE1 (1H)
	Purple	Negative Signal	Ground
	Clear	Shield	G
<b>257</b>	Black	Excitation	EX2
	Red	Positive Signal	SE2 (1L)
	White	Negative Signal	Ground
	Clear	Shield	G

## 5. Description of Measurement

The 257 probe is measured with an AC Half Bridge measurement followed by a sensor resistance calculation.

This section will distinguish between Edlog dataloggers and CRBasic dataloggers. Edlog dataloggers include the 21X, CR10(X), CR510, CR23X, and CR7. CRBasic dataloggers refer to the CR1000.

### 5.1 Edlog Dataloggers

#### 5.1.1 Program Instruction 5

Instruction 5, AC Half Bridge, is used to excite and measure the 257. Recommended excitation voltages and input ranges for Edlog dataloggers are listed in Table 5-1.

<b>TABLE 5-1. Excitation and Voltage Ranges for Edlog Dataloggers</b>			
<b>Datalogger</b>	<b>mV excitation</b>	<b>Range Code</b>	<b>Full Scale Range</b>
21X	500	14	± 500 mV
CR10(X)	250	14	± 250 mV
CR510	250	14	± 250 mV
CR23X	200	13	± 200 mV
CR7	500	16	± 500 mV

### 5.1.2 Program Instruction 59

Instruction 59, Bridge Transform, is used to output sensor resistance ( $R_s$ ). The instruction takes the AC Half Bridge output ( $V_s/V_x$ ) and computes the sensor resistance as follows:

$$R_s = R_1 \left( \frac{X}{(1 - X)} \right)$$

Where  $X = V_s/V_x$  (output from Instruction 5).

A multiplier of 1 should be used to output sensor resistance ( $R_s$ ) in terms of  $k\Omega$ .

## 5.2 CRBasic Dataloggers

### 5.2.1 BRHalf instruction

The CR1000 uses the BRHalf instruction with the RevEx argument set to True to excite and measure the 257.

Table 5-2 shows the excitation and voltage ranges for the CR1000 dataloggers.

<b>TABLE 5-2. Excitation and Voltage Ranges for CRBasic Dataloggers</b>		
<b>Datalogger</b>	<b>mV excitation</b>	<b>Full Scale Range</b>
CR1000	250	± 250mV

## 5.3 Calculate Soil Water Potential

The datalogger can calculate soil water potential (kPa) from the sensor resistance ( $R_s$ ) and soil temperature ( $T_s$ ). See Table 5-3.

The need for a precise soil temperature measurement should not be over emphasized. Soil temperatures vary widely where placement is shallow and solar radiation impinges on the soil surface. A soil temperature measurement may be needed in such situations, particularly in research applications. Many applications, however, require deep placement (12 to

25 cm) in soils shaded by a crop canopy. A common practice is to assume the air temperature at sunrise will be close to what the soil temperature will be for the day.

### 5.3.1 Linear Relationship

For applications where soil water potential is in the range of 0 to 200 kPa, water potential and temperature responses of the 257 can be assumed to be linear (measurements beyond 125 kPa have not been verified, but work in practice).

The following equation normalizes the resistance measurement to 21 °C.

$$R_{21} = \frac{R_s}{1 - (0.018 * dT)}$$

where

$R_{21}$  = resistance at 21 °C

$R_s$  = the measured resistance

$dT = T_s - 21$

$T_s$  = soil temperature

Water potential is then calculated from  $R_{21}$  with the relationship,

$$SWP = 7.407 * R_{21} - 3.704$$

where SWP is soil water potential in kPa

### 5.3.2 Non-Linear Relationship

For more precise work, calibration and temperature compensation in the range of 10 to 100 kPa has been refined by Thompson and Armstrong (1987), as defined in the non-linear equation,

$$SWP = \frac{R_s}{0.01306[1.062(34.21 - T_s + 0.01060T_s^2) - R_s]}$$

where SWP is soil water potential in kPa

<b>TABLE 5-3. Comparison of Estimated Soil Water Potential and <math>R_s</math> at 21 °C</b>		
<b>kPa (Non-Linear Equation)</b>	<b>kPa (Linear Equation)</b>	<b>(<math>R_s</math>) kOhms</b>
	3.7	1.00
9	11	2.00
14	18	3.00
20	26	4.00
27	33	5.00
35	41	6.00
45	48	7.00
56	56	8.00
69	63	9.00
85	70	10.00
105	78	11.00
	85	12.00
	92	13.00
	99	14.00
	107	15.00
	115	16.00
	122	17.00
	129	18.00
	144	20.00
	159	22.00
	174	24.00
	188	26.00
	199	27.50

## 6. Example Programs

This section is for users who write their own datalogger programs. A datalogger program to measure the 257 can be created using Campbell Scientific's Short Cut Program Builder software (SCWin). You do not need to read this section to use Short Cut.

---

**NOTE**

Short Cut requires that you add a soil temperature sensor before adding a 257 probe. This is needed because there is a temperature correction factor in the equations that convert sensor resistance.

---

### 6.1 Program Examples — CR10X

#### 6.1.1 Program Example — CR10X Linear Relationship (0 to 200 kPa)

The following example demonstrates the programming used to measure the resistance ( $k\Omega$ ) of one 257 sensor with the CR10X datalogger. The linear relationship between sensor resistance and water potential in the 0 to 200

kPa range is used. Sensor wiring for this example is shown in Table 4-1.  
Voltage range codes for other Edlog dataloggers are shown in Table 5-1.

```

;{CR10X}
*Table 1 Program
  01: 1.0000      Execution Interval (seconds)

;Measure soil temp. with 107 probe
1: Temp (107) (P11)
  1: 1           Reps
  2: 1           SE Channel
  3: 1           Excite all reps w/E1
  4: 1           Loc [ Tsoil_C ]
  5: 1.0         Multiplier
  6: 0.0         Offset

;Measure 257 block resistance
2: AC Half Bridge (P5)
  1: 1           Reps
  2: 14          250 mV Fast Range
  3: 2           SE Channel
  4: 2           Excite all reps w/Exchan 2
  5: 250         mV Excitation
  6: 2           Loc [ kOhms ]
  7: 1           Multiplier
  8: 0           Offset

;Convert voltage reading to kOhms
3: BR Transform Rf[X/(1-X)] (P59)
  1: 1           Reps
  2: 2           Loc [ kOhms ]
  3: 1           Multiplier (Rf)

;Calculate dT = T -21
4: Z=X+F (P34)
  1: 1           X Loc [ Tsoil_C ]
  2: -21         F
  3: 4           Z Loc [ CorFactr ]

;Calculate (0.018 * dT)
5: Z=X*F (P37)
  1: 4           X Loc [ CorFactr ]
  2: 0.018       F
  3: 4           Z Loc [ CorFactr ]

;Calculate (1 - (0.018 * dT))
6: Z=X+F (P34)
  1: 4           X Loc [ CorFactr ]
  2: -1          F
  3: 4           Z Loc [ CorFactr ]

7: Z=X*F (P37)
  1: 4           X Loc [ CorFactr ]
  2: -1          F
  3: 4           Z Loc [ CorFactr ]

```

```

;Apply Temperature correction and sensor
;Calibration to kOhm measurements.
;See linear equation in 5.3.1

;Temperature correct kOhms
8: Z=X/Y (P38)
  1: 2          X Loc [ kOhms  ]
  2: 4          Y Loc [ CorFactr ]
  3: 3          Z Loc [ WP_kPa  ]

;Apply calibration slope and offset
9: Z=X*F (P37)
  1: 3          X Loc [ WP_kPa  ]
  2: 7.407      F
  3: 3          Z Loc [ WP_kPa  ]

10: Z=X+F (P34)
  1: 3          X Loc [ WP_kPa  ]
  2: -3.704     F
  3: 3          Z Loc [ WP_kPa  ]

;Send measurements to final storage hourly
11: If time is (P92)
  1: 0          Minutes (Seconds --) into a
  2: 60         Interval (same units as above)
  3: 10        Set Output Flag High (Flag 0)

12: Set Active Storage Area (P80)
  1: 1          Final Storage Area 1
  2: 60         Array ID

13: Real Time (P77)
  1: 1220       Year,Day,Hour/Minute (midnight = 2400)

14: Average (P71)
  1: 1          Reps
  2: 1          Loc [ Tsoil_C  ]

15: Sample (P70)
  1: 1          Reps
  2: 3          Loc [ WP_kPa  ]

```

### 6.1.2 Program Example — CR10X Non-Linear Relationship (10 to 100 kPa)

The following example demonstrates the programming used to measure the resistance (kΩ) of one 257 sensor with the CR10X datalogger. The non-linear relationship between sensor resistance and water potential in the 10 to 100 kPa range is used. Sensor wiring for this example is shown in Table 4-1. Voltage range codes for other Edlog dataloggers are shown in Table 5-1.

```

;{CR10X}
*Table 1 Program
 01: 1.0000      Execution Interval (seconds)

;Measure soil temperature with 107 probe
1: Temp (107) (P11)
 1: 1           Reps
 2: 1           SE Channel
 3: 1           Excite all reps w/E1
 4: 1           Loc [ Tsoil_C ]
 5: 1.0         Multiplier
 6: 0.0         Offset

;Measure 257 block resistance
2: AC Half Bridge (P5)
 1: 1           Reps
 2: 14          250 mV Fast Range
 3: 2           SE Channel
 4: 2           Excite all reps w/Exchan 2
 5: 250         mV Excitation
 6: 2           Loc [ kOhms ]
 7: 1           Multiplier
 8: 0           Offset

;Convert voltage reading to kOhms
3: BR Transform Rf[X/(1-X)] (P59)
 1: 1           Reps
 2: 2           Loc [ kOhms ]
 3: 1           Multiplier (Rf)

;Calculate Tsoil^2
4: Z=X*Y (P36)
 1: 1           X Loc [ Tsoil_C ]
 2: 1           Y Loc [ Tsoil_C ]
 3: 3           Z Loc [ WP_kPa ]

;SWP = Tsoil^2 * 0.0106
5: Z=X*F (P37)
 1: 3           X Loc [ WP_kPa ]
 2: 0.0106     F
 3: 3           Z Loc [ WP_kPa ]

6: Z=F x 10^n (P30)
 1: 34.21      F
 2: 0          n, Exponent of 10
 3: 4          Z Loc [ CorFactr ]

;SWP = 34.21 - Tsoil
7: Z=X-Y (P35)
 1: 4          X Loc [ CorFactr ]
 2: 1          Y Loc [ Tsoil_C ]
 3: 4          Z Loc [ CorFactr ]

```

```

;SWPcalc = [(34.21 - Tsoil) + (Tsoil^2 * 0.01060)]
8: Z=X+Y (P33)
  1: 3      X Loc [ WP_kPa ]
  2: 4      Y Loc [ CorFactr ]
  3: 3      Z Loc [ WP_kPa ]

;SWP = 1.062[SWPcalc]
9: Z=X*F (P37)
  1: 3      X Loc [ WP_kPa ]
  2: 1.062  F
  3: 3      Z Loc [ WP_kPa ]

;SWP = SWPcalc - Rs
10: Z=X-Y (P35)
  1: 3      X Loc [ WP_kPa ]
  2: 2      Y Loc [ kOhms ]
  3: 3      Z Loc [ WP_kPa ]

;SWP = 0.01306 * SWPcalc
11: Z=X*F (P37)
  1: 3      X Loc [ WP_kPa ]
  2: 0.0130 F
  3: 3      Z Loc [ WP_kPa ]

;SWP = Rs/SWPcalc
12: Z=X/Y (P38)
  1: 2      X Loc [ kOhms ]
  2: 3      Y Loc [ WP_kPa ]
  3: 3      Z Loc [ WP_kPa ]

;Send measurements to final storage hourly
13: If time is (P92)
  1: 0      Minutes (Seconds --) into a
  2: 60     Interval (same units as above)
  3: 10     Set Output Flag High (Flag 0)

14: Set Active Storage Area (P80)
  1: 1      Final Storage Area 1
  2: 60     Array ID

15: Real Time (P77)
  1: 1220   Year,Day,Hour/Minute (midnight = 2400)

16: Average (P71)
  1: 1      Reps
  2: 1      Loc [ Tsoil_C ]

17: Sample (P70)
  1: 1      Reps
  2: 3      Loc [ WP_kPa ]

```

## 6.2 Program Example — CR1000

The following example demonstrates the programming used to measure the resistance (k $\Omega$ ) of one 257 sensor with the CR1000 datalogger. The equations for both the linear and non-linear relationships between sensor resistance and water potential are shown. Sensor wiring for this example is shown in Table 4-2.

```
'CR1000

'Declare Variables and Units
Public T107_C
Public kOhms
Public WP_kPa

Units T107_C=Deg C
Units kOhms=kOhms
Units WP_kPa=kPa

'Define Data Tables
DataTable(T257,True,-1)
  DataInterval(0,60,Min,10)
  Average(1,T107_C,FP2,False)
  Sample(1,WP_kPa,FP2)
EndTable

'Main Program
BeginProg
  Scan(1,Sec,1,0)
  '107 Temperature Probe measurement T107_C:
  Therm107(T107_C,1,1,1,0,_60Hz,1.0,0.0)
  '257 Soil Moisture Sensor measurements kOhms and WP_kPa:
  BrHalf(kOhms,1,mV250,2,Vx2,1,250,True,0,250,1,0)
  kOhms=kOhms/(1-kOhms)
  'Equation for linear (0 to 200 kPa) relationship
  WP_kPa=0.07407*kOhms/(1-0.018*(T107_C-21))-0.03704

  'For nonlinear (10 to 100 kPa) relationship, use the following equation
  'WP_kPa=kOhms/(0.01306*(1.062*(34.21-T107_C+0.01060*T107_C^2)-kOhms))*0.01

  WP_kPa=WP_kPa*100
  'Call Data Tables and Store Data
  CallTable(T257)
NextScan
EndProg
```

## 7. Interpreting Results

As a general guide, 257 measurements indicate soil moisture as follows:

- 0 to 10 kPa = Saturated soil
- 10 to 20 kPa = Soil is adequately wet (except coarse sands, which are beginning to lose water).
- 20 to 60 kPa = Usual range for irrigation (except heavy clay).
- 60 to 100 kPa = Usual range for irrigation for heavy clay soils.
- 100 to 200 kPa = Soil is becoming dangerously dry for maximum production.

## 8. Troubleshooting

To test the sensor, submerge it in water. Measurements should be from -3 to +3 kPa. Let the sensor dry for 30 to 48 hours. You should see the reading increase from 0 to 15,000 + kPa. Put the sensor back in the water. The reading should run right back down to zero in 1 to 2 minutes. If the sensor passes these tests, consider the following:

1. Sensor may not have a snug fit in the soil. This usually happens when an oversized access hole has been used and the backfilling of the area around the sensor is not complete.
2. Sensor is not in an active portion of the root system, or the irrigation is not reaching the sensor area. This can happen if the sensor is sitting on top of a rock or below a hard pan which may impede water movement. Re-installing the sensor usually solves this problem.
3. When the soil dries out to the point where you are seeing readings higher than 80 kPa, the contact between soil and sensor can be lost because the soil may start to shrink away from the sensor. An irrigation which only results in a partial rewetting of the soil will not fully rewet the sensor, which can result in continued high readings from the 257. Full rewetting of the soil and sensor usually restores soil to sensor contact. This is most often seen in the heavier soils and during peak crop water demand when irrigation may not be fully adequate. The plotting of readings on a chart is most useful in getting a good picture of this sort of behavior.

## 9. Reference

Thompson, S.J. and C.F. Armstrong, Calibration of the Watermark Model 200 Soil Moisture Sensor, Applied Engineering in Agriculture, Vol. 3, No. 2, pp. 186-189, 1987.

Parts of this manual were contributed by Irrrometer Company, Inc., manufacturer of the Watermark 200.





## **Campbell Scientific Companies**

---

### **Campbell Scientific, Inc. (CSI)**

815 West 1800 North  
Logan, Utah 84321  
UNITED STATES  
[www.campbellsci.com](http://www.campbellsci.com)  
[info@campbellsci.com](mailto:info@campbellsci.com)

### **Campbell Scientific Africa Pty. Ltd. (CSAf)**

PO Box 2450  
Somerset West 7129  
SOUTH AFRICA  
[www.csafrica.co.za](http://www.csafrica.co.za)  
[sales@csafrica.co.za](mailto:sales@csafrica.co.za)

### **Campbell Scientific Australia Pty. Ltd. (CSA)**

PO Box 444  
Thuringowa Central  
QLD 4812 AUSTRALIA  
[www.campbellsci.com.au](http://www.campbellsci.com.au)  
[info@campbellsci.com.au](mailto:info@campbellsci.com.au)

### **Campbell Scientific do Brazil Ltda. (CSB)**

Rua Luisa Crapsi Orsi, 15 Butantã  
CEP: 005543-000 São Paulo SP BRAZIL  
[www.campbellsci.com.br](http://www.campbellsci.com.br)  
[suporte@campbellsci.com.br](mailto:suporte@campbellsci.com.br)

### **Campbell Scientific Canada Corp. (CSC)**

11564 - 149th Street NW  
Edmonton, Alberta T5M 1W7  
CANADA  
[www.campbellsci.ca](http://www.campbellsci.ca)  
[dataloggers@campbellsci.ca](mailto:dataloggers@campbellsci.ca)

### **Campbell Scientific Ltd. (CSL)**

Campbell Park  
80 Hathern Road  
Shepshed, Loughborough LE12 9GX  
UNITED KINGDOM  
[www.campbellsci.co.uk](http://www.campbellsci.co.uk)  
[sales@campbellsci.co.uk](mailto:sales@campbellsci.co.uk)

### **Campbell Scientific Ltd. (France)**

Miniparc du Verger - Bat. H  
1, rue de Terre Neuve - Les Ulis  
91967 COURTABOEUF CEDEX  
FRANCE  
[www.campbellsci.fr](http://www.campbellsci.fr)  
[campbell.scientific@wanadoo.fr](mailto:campbell.scientific@wanadoo.fr)

### **Campbell Scientific Spain, S. L.**

Psg. Font 14, local 8  
08013 Barcelona  
SPAIN  
[www.campbellsci.es](http://www.campbellsci.es)  
[info@campbellsci.es](mailto:info@campbellsci.es)